
Bibliothèque Ada avec GNAT

1) Principes

GNAT nous permet d'utiliser par défaut la bibliothèque standard Ada  et 
d'autres avec quelques options supplémentaires (voir son utilisation sur Blady). 
Nous allons aborder dans ce tutoriel comment créer notre propre bibliothèque 
Ada pour une utilisation avec un programme principal en Ada puis aussi avec un 
programme principal en C ou en Objective-C.

2) Création d'une bibliothèque Ada statique indépendante

Nous souhaitons construire une bibliothèque indépendante (pouvant être 
installée et utilisée ailleurs qu'à l'endroit où elle a été construite) et à l'édition 
des liens statique. Pour cela nous créons deux projets GPR. Un premier projet 
pour construire la bibliothèque et un second pour l'utiliser.

Le premier projet va compiler tout les sources du répertoire "src" dans le 
répertoire "obj" et construire la bibliothèque dans le répertoire "lib". Ces trois 
répertoires doivent exister. Il s'agit d'un projet comportant les options 
classiques avec néanmoins l'option "for Main use" remplacée par deux nouvelles 
options "for Library_Name use" et "for Library_Dir use" :

• Library_Name : nom de base pour la construction du nom complet par 
exemple "bibli" donnera "libbibli.a"

• Library_Dir : répertoire résultat de la construction de la bibliothèque

Options facultatives :

• Library_Kind : nature de la bibliothèque statique (par défaut) ou 
dynamique ("static", "dynamic" or "relocatable")

• Library_Options : options (finales) pour l'édition des liens
• Leading_Library_Options : options (initiales) pour l'édition des liens
• Linker.Linker_Options: options à passer à l'édition des liens du 

programme utilisateur



Le mot-clé library avant project est optionnel mais recommandé pour une 
meilleure lisibilité sans équivoque :

$ cat > bibli_build.gpr
library project Bibli_Build is
   for Source_Dirs use ("src");
   for Object_Dir use "obj";
   for Library_Name use "bibli";
   for Library_Dir use "lib";
end Bibli_Build;

Notre bibliothèque "bibli" peut alors contenir plusieurs paquetages. Prenons 
l'exemple d'un paquetage Bibli1 :

$ cat > src/bibli1.ads
package Bibli1 is
procedure Affiche;
end;
$ cat > src/bibli1.adb
with Ada.Text_IO;
use Ada.Text_IO;
package body Bibli1 is
   procedure Affiche is
   begin
      Put_Line("Bibli1.Affiche");
   end Affiche;
end Bibli1;

Construisons alors la bibliothèque avec la commande suivante :

$ gnatmake -P bibli_build.gpr 
gcc -c -I- -gnatA ./src/bibli1.adb
building static library for project bibli_build
ar cr ./lib/libbibli.a ./obj/bibli1.o
ranlib ./lib/libbibli.a

Le deuxième projet va contenir toute les informations nécessaires à 
l'utilisation de la bibliothèque. Nous mettons ce projet dans le répertoire "lib/
gnat" et les spécifications ainsi que les corps des paquetages génériques 
(spécifique à GNAT) dans le répertoire "include" :

$ mkdir lib/gnat
$ cat > lib/gnat/bibli.gpr
library project Bibli is
   for Source_Dirs use ("../../include");
   for Library_Name use "bibli";
   for Library_Dir use "..";



   for Externally_Built use "true";
end Bibli;
$ mkdir include
$ cp -p src/*.ads include

Les répertoires "Source_Dirs" et "Library_Dir" sont relatifs à 
l'emplacement de bibli.gpr. L'option "for Externally_Built use "true";" permet de 
spécifier que la bibliothèque est indépendante.

Notre programme d'essai est construit de manière classique avec le 
projet :

$ cat > essais.gpr 
with "./lib/gnat/bibli.gpr";
project Essais is
   for Main use ("hello4.adb");
end Essais;

Construisons et exécutons notre programme principal :

$ cat > hello4.adb
with Bibli1;
procedure Hello4 is
begin
   Bibli1.Affiche;
end Hello4;
$ gnatmake -P essais.gpr 
gcc -c -I- -gnatA hello4.adb
gnatbind -I- -x hello4.ali
gnatlink hello4.ali lib/libbibli.a -Wl,-rpath,/usr/local/gnat-2012/lib/gcc/x86_64-apple-
darwin10.8.0/4.5.4/adalib/ -o hello4
$ ./hello4 
Bibli1.Affiche

Voilà donc notre première bibliothèque Ada construite avec GNAT.



3) Création d'une bibliothèque Ada statique indépendante externe

Nous souhaitons construire une bibliothèque indépendante (pouvant être 
installée et utilisée ailleurs qu'à l'endroit où elle a été construite), à l'édition 
des liens statique et pouvant être utilisée par un autre langage comme le C. Pour 
cela nous modifions notre premier projet GPR en ajoutant l'option "for 
Library_Interface use" :

• Library_Interface : liste des spécifications Ada exportées

Notre projet devient :

$ cat bibli_build.gpr 
library project Bibli_Build is
   for Source_Dirs use ("src");
   for Object_Dir use "obj";
   for Library_Name use "bibli";
   for Library_Dir use "lib";
   for Library_Interface use ("Bibli1");
end Bibli_Build;

Construisons alors la bibliothèque avec la commande suivante :

$ gnatmake -P bibli_build.gpr
gcc -c -g -I- -gnatA ./src/bibli1.adb
gnatbind -n -o b~bibli.adb -Lbibli -a ...
gcc -c -gnatws b~bibli.adb -fPIC -g ...
building static library for project bibli_build
ar cr ./lib/libbibli.a ./obj/bibli1.o ...
ranlib ./lib/libbibli.a

Comme notre bibliothèque est devenue externe, le mécanisme GNAT 
d'initialisation des bibliothèques n'est plus automatique. Nous devons l'ajouter 
dans notre programme principal. Modifions, construisons et exécutons notre 
programme principal :

$ cat hello4.adb
with Bibli1;
procedure Hello4 is
   procedure bibliinit;
   pragma Import (C, bibliinit, "bibliinit");
   procedure biblifinal;
   pragma Import (C, biblifinal, "biblifinal");
begin
BibliInit;
Bibli1.Affiche;



BibliFinal;
end;

$ gnatmake -P essais.gpr
gcc -c -g -g -I- -gnatA ./hello4.adb
gnatbind -E -I- -x ./hello4.ali
gnatlink ./hello4.ali -g ./lib/libbibli.a -Wl,-rpath,/usr/local/gnat-2012/lib/gcc/x86_64-
apple-darwin10.8.0/4.5.4/adalibb/ -o ./hello4
$ ./hello4 
Bibli1.Affiche

Il nous reste maintenant une dernière modification à réaliser pour 
permettre à du code en langage C à appeler notre sous-programme d'affichage 
en Ada. Il nous faut l'exporter :

$ cat src/bibli1.ads
package Bibli1 is
procedure Affiche;
pragma Export (C, Affiche, "affiche");
end;
$ cp -p src/*.ads include

Nous ajoutons également le fichier d'interface C habituel aux 
programmeurs en C :

$ cat > include/bibli1.h 
/* the library elaboration procedure */
extern void bibliinit (void);
/* the library finalization procedure */
extern void biblifinal (void);
/* the interface exported by the library */
extern void affiche (void);

Construisons alors la bibliothèque avec la commande suivante :

$ gnatmake -P bibli_build.gpr
gcc -c -g -I- -gnatA ./src/bibli1.adb
gnatbind -n -o b~bibli.adb -Lbibli -a ...
gcc -c -gnatws b~bibli.adb -fPIC -g ...

building static library for project bibli_build
ar cr ./lib/libbibli.a ./obj/bibli1.o ...
ranlib ./lib/libbibli.a



Nous pouvons alors écrire notre programme principal en C, le compiler et 
l'exécuter :

$ cat > main.c
#include "bibli1.h"
int main (void)
{
   /* En premier, initialisation de la bibliothèque */
   bibliinit ();
   /* Programme principal utilisant les entités exportées */
   affiche ();
   /* En dernier, finalisation de la bibliothèque */
   biblifinal ();
   return 0;
}
$ gcc main.c -Iinclude -Llib -lbibli /usr/local/gnat/lib/gcc/x86_64-apple-
darwin10.8.0/4.5.4/adalib/libgnat.a
$ ./a.out 
Bibli1.Affiche

4) Utilisation de notre bibliothèque Ada avec XCode

Le compilateur GNAT GPL ne compile pas par défaut le langage Objective-
C. Nous allons utiliser XCode qui embarque le compilateur Objective-C.

Lancez Xcode et cliques sur "Create a new Xcode project" ou sur le menu 
"File->New->Project..." puis dans le panneau ouvert sélectionnez "OS X-
>Application->Command Line Tool" et cliquez sur "Next". Saisir "Essai01" comme 
"Product Name" et sélectionnez le type "Foundation" puis cliquez sur "Next". 
Sélectionnez un répertoire où XCode va créer un nouveau répertoire "Essai01" 
et cochez la case "Source Control" puis cliquez sur "Create", la fenêtre de notre 
projet s'affiche.

Nous allons créez notre propre classe Objective-C. Sélectionez le menu 
"File->New->File..." puis dans le panneau ouvert sélectionnez "OS X->Objective)C 
class" puis cliquez sur "Next". Saisir "Classe01" comme champ "Class", laissez le 
champ "Subclass of" positionné sur "NSObject" puis cliquez sur "Next". Vérifiez 
que "Essai01" est bien coché dans le tableau "Targets" puis cliquez sur "Create".

Nous allons ajouter la déclaration des variables et méthodes d'instances 
dans le fichier d'interface Classe01.h (ajouts en gras) :

#import <Foundation/Foundation.h>

@interface Classe01 : NSObject
{
    int n;
    double factorielle;
}



@property int n;
@property double factorielle;

- (void) calculFactorielle;

@end

Vous remarquez que nous utilisons la forme simplifiée des propriétés. 
Ajoutez également l'implémentation des méthodes avec la même forme 
simplifiée pour les variables ainsi que la fonction de calcul de la factorielle dans 
Classe01.m :

#import "Classe01.h"

@implementation Classe01

@synthesize n, factorielle;

float fact (int n) {
    if (n <= 1) return 1.0;
    return (float) n * fact (n - 1);
}

- (void) calculFactorielle {
    factorielle = fact(n);
}

@end

Il ne nous reste plus qu'à créer et utiliser notre classe dans le programme 
principal main.m :

#import <Foundation/Foundation.h>
#import "Classe01.h"

int main(int argc, const char * argv[])
{

    @autoreleasepool {
        
        // insert code here...
        Classe01 *calcul = [[Classe01 alloc] init];
        [calcul setN:20];
        [calcul calculFactorielle];
        NSLog(@"Factorielle %d : %f", [calcul n], [calcul 
factorielle]);
        
    }
    return 0;
}



Cliquez sur "Run" ou sélectionner le menu "Product->Run", le résultat ne se 
fait pas attendre dans la zone "All Output" :

2013-03-17 11:35:41.122 Essai01[22369:303] Factorielle 20 : 
2432902023163674624.000000

La mécanique, assez artificielle, j'en conviens, de notre classe de calcul de 
factorielle fonctionne, si vous avez coché la case "Source Control", vous pouvez 
référencer le code en sélectionnant les trois sources main.m, Classe01.h, 
Classe01.m puis le menu "File->Source Control->Commit...". Ajoutez un 
commentaire au niveau de la zone "Enter commit message here" puis cliquez sur 
le bouton "Commit Files". Le référencement est visible dans la fenêtre 
accessible par le menu "Window->Organizer->Repositories".

Nous allons alors utiliser notre bibliothèque en Ada "Bibli". Vous pouvez 
copiez les répertoires "include", "lib", "obj", "src" et le projet "bibli_build.gpr" 
dans le répertoire "Essai01". Ce n'est pas obligatoire, ce sera plus simple dans la 
configuration du projet XCode. Pour cela, cliquez sur le projet "Essai01" dans le 
navigateur de projets. Dans la zone du projet qui vient de s'afficher, cliquez sur 
le projet "Essai01" puis "Build Settings". Recherchez la section "Search Paths" 
puis la sous-section "Header Search Paths", déployez la avec le triangle noir puis 
cliquez sur le signe + en face de "Debug" pour saisir juste "include" ou le chemin 
complet. Faites de même avec "Library Search Paths" pour saisir juste "lib"  ou 
le chemin complet en face de "Debug". Recherchez la section "Linking" puis la 
sous-section "Other Linker Flags", déployez la avec le triangle noir puis cliquez 
sur le signe + en face de "Debug" pour saisir : 
"-libibli /usr/local/gnat/lib/gcc/x86_64-apple-darwin10.8.0/4.5.4/adalib/
libgnat.a".

Voilà notre configuration est prête à utiliser la bibliothèque Ada. Avec 
XCode, modifiez maintenant le programme principal pour ajouter la référence à 
"bibli1.h" les instructions d'initialisation, d'utilisation et de finalisation de la 
bibliothèque Ada Bibli :

#import <Foundation/Foundation.h>
#import "Classe01.h"
#include "bibli1.h"

int main(int argc, const char * argv[])
{

    @autoreleasepool {
        
        // insert code here...
        /* First, elaborate the library before using it */
        bibliinit ();



        Classe01 *calcul = [[Classe01 alloc] init];
        [calcul setN:20];
        [calcul calculFactorielle];
        NSLog(@"Factorielle %d : %f", [calcul n], [calcul 
factorielle]);

        /* Main program, using the library exported entities */
        affiche ();
        /* Library finalization at the end of the program */
        biblifinal ();

    }
    return 0;
}

Cliquez sur Run :

Factorielle 20 : 2432902023163674624.000000
Bibli1.Affiche

Ça fonctionne, profitons en pour référencer les sources Ada avec GIT  GUI 
(voir sur Blady) puisque XCode ne les voit pas par défaut.

5) Méthodes d'instance et de méthodes de classe

Nous allons ajouter une méthode de classe identifiée par le signe + dans 
Classe01.h :

#import <Foundation/Foundation.h>

@interface Classe01 : NSObject
{
    int n;
    float factorielle;
}

@property int n;
@property float factorielle;

+ (void) quiSuisJe;
- (void) calculFactorielle;

@end



et Classe01.m :

#import "Classe01.h"

@implementation Classe01

@synthesize n, factorielle;

float fact (int n) {
    if (n <= 1) return 1.0;
    return (float) n * fact (n - 1);
}

+ (void) quiSuisJe {
    NSLog(@"Je suis la classe %@.", [Classe01 class]);
}

- (void) calculFactorielle {
    factorielle = fact(n);
}

@end

Puis ajoutons les déclarations dans le Bibli1.ads :

with System;
package Bibli1 is
   type id is new System.Address;
   procedure Affiche;
   pragma Export (C, Affiche, "affiche");
   procedure CalcInstance (This : id);
   pragma Export (C, CalcInstance, "calcinst");
   procedure CalcClasse;
   pragma Export (C, CalcClasse, "calcclasse");
end Bibli1;

ainsi que dans le fichier d'interface C bibli1.h :

/* the library elaboration procedure */
extern void bibliinit (void);
/* the library finalization procedure */
extern void biblifinal (void);
/* the interface exported by the library */
extern void affiche (void);
extern void calcinst (id this);
extern void calcclasse (void);



ainsi que le code dans Bibli1.adb :

with Ada.Text_IO;          use Ada.Text_IO;
with Interfaces.C.Strings; use Interfaces.C.Strings;
with System.Address_Image;
package body Bibli1 is
   procedure Affiche is
   begin
      Put_Line ("Bibli1.Affiche");
   end Affiche;
   type SEL is new System.Address;
   type Class is new System.Address;
   type Method is new System.Address;
   function objc_getClass (name : chars_ptr) return Class;
   pragma Import (C, objc_getClass, "objc_getClass");
   function class_getName (cls : Class) return chars_ptr;
   pragma Import (C, class_getName, "class_getName");
   function object_getClassName (obj : id) return chars_ptr;
   pragma Import (C, object_getClassName, "object_getClassName");
   function sel_registerName (str : chars_ptr) return SEL;
   pragma Import (C, sel_registerName, "sel_registerName");
   function sel_getName (aSelector : SEL) return chars_ptr;
   pragma Import (C, sel_getName, "sel_getName");
   function objc_msgSend (theReceiver : id; theSelector : SEL) return id;
   pragma Import (C, objc_msgSend, "objc_msgSend");
   procedure CalcInstance (This : id) is
      Dum_ID : id;
      pragma Unreferenced (Dum_ID);
   begin
      Put_Line (System.Address_Image (System.Address (This)));
      Put_Line (Value (object_getClassName (This)));
      Put_Line (Value (sel_getName (sel_registerName (New_String 
("calculFactorielle")))));
      Dum_ID := objc_msgSend (This, sel_registerName (New_String 
("calculFactorielle")));
   end CalcInstance;

   procedure CalcClasse is
      cl01   : constant Class := objc_getClass (New_String ("Classe01"));
      swai   : constant SEL   := sel_registerName (New_String ("quiSuisJe"));
      Dum_ID : id;
      pragma Unreferenced (Dum_ID);
   begin
      Put_Line ("Code Ada :");
      Put_Line (Value (class_getName (cl01)));
      Put_Line (Value (sel_getName (swai)));
      Dum_ID := objc_msgSend (id (cl01), swai);
   end CalcClasse;
end Bibli1;



Heureusement le Runtime Objective-C est composé de fonction C, nous 
pouvons donc les importer à loisir dans notre code Ada :

- sel_registerName : renvoie un index sur la méthode désignée
- sel_getName : renvoie le nom de la méthode
- objc_getClass : renvoie un pointeur sur la classe désignée
- class_getName : renvoie le nom de la classe
- class_getClassMethod : renvoie un pointeur sur la méthode de classe
- method_getImplementation : renvoie un pointeur
- object_getClassName : renvoie le nom de classe de l'instance désignée
- objc_msgSend : envoie un message à une méthode d'une instance

L'ensemble des fonctions du runtime est décrit dans la documentation du 
développeur :

https://developer.apple.com/library/mac/documentation/Cocoa/Reference/
ObjCRuntimeRef/Reference/reference.html#//apple_ref/doc/uid/TP40001418

Compilez maintenant la bibliothèque modifiée :

$ gnatmake -P bibli_build.gpr
gnatbind -n -o b~bibli.adb -Lbibli -a ...
gcc -c -gnatws b~bibli.adb -fPIC -g ...

building static library for project bibli_build
ar cr ./lib/libbibli.a ./obj/bibli1.o ...
ranlib ./lib/libbibli.a

Avec XCode, il ne nous reste plus qu'à appeler nos nouvelles procédure Ada 
dans le programme principal :

#import <Foundation/Foundation.h>
#import "Classe01.h"
#include "bibli1.h"

int main(int argc, const char * argv[])
{

    @autoreleasepool {
        
        // insert code here...
        /* First, elaborate the library before using it */
        bibliinit ();

        Classe01 *calcul = [[Classe01 alloc] init];
        [Classe01 quiSuisJe];



        [calcul setN:20];
        [calcul calculFactorielle];
        NSLog(@"Factorielle Obj-C %d : %f", [calcul n], [calcul 
factorielle]);

        /* Main program, using the library exported entities */
        affiche ();
        [calcul setN:10];
        calcinst (calcul);
        NSLog(@"Factorielle Ada %d : %f", [calcul n], [calcul 
factorielle]);
        calcclasse ();
        /* Library finalization at the end of the program */
        biblifinal ();

    }
    return 0;
}

La procédure CalcInst appelle la méthode d'instance "calculFactorielle" et 
CalcClasse appelle la méthode de classe "quiSuisJe". Cliquez sur Run :

Essai01[51724:303] Code Objective-C :
Essai01[51724:303] Je suis la classe Classe01.
Essai01[51724:303] Factorielle Obj-C 20 : 2432902023163674624.000000
Bibli1.Affiche
00000001001082E0
Classe01
calculFactorielle
Essai01[51724:303] Factorielle Ada 10 : 3628800.000000
Code Ada :
Classe01
quiSuisJe
Essai01[51724:303] Je suis la classe Classe01.

6) Allocation d'une instance de classe

Reprenons notre bibli1 et ajoutons une procédure de création et 
d'utilisation complète de la classe de calcul de factorielles en Ada. Ajoutons 
alors la déclaration dans le Bibli1.ads :

with System;
with Interfaces.C;
package Bibli1 is
   type id is new System.Address;
   procedure Affiche;
   pragma Export (C, Affiche, "affiche");
   procedure CalcInstance (This : id);



   pragma Export (C, CalcInstance, "calcinst");
   procedure CalcClasse;
   pragma Export (C, CalcClasse, "calcclasse");
   procedure CalcFact (N : Interfaces.C.int);
   pragma Export (C, CalcFact, "calcfact");
end Bibli1;

ainsi que dans le fichier d'interface C bibli1.h :

/* the library elaboration procedure */
extern void bibliinit (void);
/* the library finalization procedure */
extern void biblifinal (void);
/* the interface exported by the library */
extern void affiche (void);
extern void calcinst (id this);
extern void calcclasse (void);
extern void calcfact (int n);

ainsi que le code dans Bibli1.adb :

with Ada.Text_IO;          use Ada.Text_IO;
with Interfaces.C.Strings; use Interfaces.C.Strings;
with System.Address_Image;
package body Bibli1 is
...
   procedure CalcFact (N : Interfaces.C.int) is
      cl01               : constant Class := objc_getClass (New_String ("Classe01"));
      NSString           : constant Class := objc_getClass (New_String 
("NSString"));
      salloc             : constant SEL   := sel_registerName (New_String ("alloc"));
      sinit              : constant SEL   := sel_registerName (New_String ("init"));
      ssetN              : constant SEL   := sel_registerName (New_String ("setN:"));
      scalculFactorielle : constant SEL   := sel_registerName (New_String 
("calculFactorielle"));
      sn                 : constant SEL   := sel_registerName (New_String ("n"));
      sfactorielle       : constant SEL   := sel_registerName (New_String 
("factorielle"));
      sinitWithCString   : constant SEL   :=
        sel_registerName (New_String ("initWithCString:encoding:"));
      calc               : id;
      str                : id;
      function objc_msgSend
        (theReceiver : id;
         theSelector : SEL;
         N           : Interfaces.C.int)
         return        id;
      function objc_msgSend (theReceiver : id; theSelector : SEL) return 
Interfaces.C.C_float;



      function objc_msgSend (theReceiver : id; theSelector : SEL) return 
Interfaces.C.int;
      function objc_msgSend
        (theReceiver : id;
         theSelector : SEL;
         S           : Interfaces.C.Strings.chars_ptr;
         Encoding    : Interfaces.C.unsigned_long)
         return        id;
      pragma Import (C, Objc_Msgsend, "objc_msgSend");
      procedure NSLogIntDouble
        (Format : id;
         Entier : Interfaces.C.int;
         Reel   : Interfaces.C.double);
      pragma Import (C, NSLogIntDouble, "NSLog");
      Dum_ID : id;
      pragma Unreferenced (Dum_ID);
      Resultat : Interfaces.C.C_float;
   begin
      calc     := objc_msgSend (id (cl01), salloc);
      calc     := objc_msgSend (calc, sinit);
      Dum_ID   := objc_msgSend (calc, ssetN, N);
      Dum_ID   := objc_msgSend (calc, scalculFactorielle);
      Resultat := objc_msgSend (calc, sfactorielle);
      str      := objc_msgSend (id (NSString), salloc);
      str      := objc_msgSend (str, sinitWithCString, New_String ("Factorielle 
%d : %f"), 1);
      NSLogIntDouble (str, objc_msgSend (calc, sn), Interfaces.C.double 
(Resultat));
   end CalcFact;
end Bibli1;

Compilez maintenant la bibliothèque modifiée :

$ gnatmake -P bibli_build.gpr
gnatbind -n -o b~bibli.adb -Lbibli -a ...
gcc -c -gnatws b~bibli.adb -fPIC -g ...

building static library for project bibli_build
ar cr ./lib/libbibli.a ./obj/bibli1.o ...
ranlib ./lib/libbibli.a



Avec XCode, il ne nous reste plus qu'à appeler notre nouvelle procédure 
Ada dans le programme principal :

int main(int argc, const char * argv[])
{

...
        calcclasse ();
        calcfact (15);
        /* Library finalization at the end of the program */
        biblifinal ();

    }
    return 0;
}

Cliquez sur Run :

Essai01[15465:303] Code Objective-C :
Essai01[15465:303] Je suis la classe Classe01.
Essai01[15465:303] Factorielle Obj-C 20 : 2432902023163674624.000000
Bibli1.Affiche
00000001001082E0
Classe01
calculFactorielle
Essai01[15465:303] Factorielle Ada 10 : 3628800.000000
Code Ada :
Classe01
quiSuisJe
Essai01[15465:303] Je suis la classe Classe01.
Essai01[15465:303] Factorielle 15 : 1307674279936.000000

Voilà notre bibliothèque crée directement en Ada des instances de classes 
Objective-C.

Pascal Pignard, février-mai 2013.


